// This Pine Script™ code is subject to the terms of the Mozilla Public License 2.0 at https://mozilla.org/MPL/2.0/
// © ChartPrime

//@version=5
indicator("Multi Deviation Scaled Moving Average [ChartPrime]", "Multi DSMA - [ChartPrime]", overlay=true)

// ---------------------------------------------------------------------------------------------------------------------}
// 𝙐𝙎𝙀𝙍 𝙄𝙉𝙋𝙐𝙏𝙎
// ---------------------------------------------------------------------------------------------------------------------{

int period          = input.int(30, title="Period")
int step            = 100 - input.int(60, "Sensitivity", minval = 0, maxval = 100,  tooltip = "The Lower input Lower sensitivity")
series float src    = hlc3

// Colors for visualization
color upper_color   = #41a1ce
color down_color    = #ce8541

// ---------------------------------------------------------------------------------------------------------------------}
// 𝙄𝙉𝘿𝙄𝘾𝘼𝙏𝙊𝙍 𝘾𝘼𝙇𝘾𝙐𝙇𝘼𝙏𝙄𝙊𝙉𝙎
// ---------------------------------------------------------------------------------------------------------------------{

// @function Calculates the Deviation Scaled Moving Average (DSMA)
// @param src (series float) Source price series
// @param period (int) Calculation period
// @returns (float) DSMA value
dsma(src, int period)=>
    // Variables initialization
    var float a1   = 0.0
    var float b1   = 0.0
    var float c1   = 0.0
    var float c2   = 0.0
    var float c3   = 0.0
    var float filt = 0.0
    var float dsma = 0.0
    var float s    = 0.0

    // Initialize variables on the first bar
    if barstate.isfirst
        pi = 3.1415926535897932
        g  = math.sqrt(2)
        // Smooth with a Super Smoother
        s  := 2 * pi / period
        a1 := math.exp(-g * pi / (0.5 * period))
        b1 := 2 * a1 * math.cos(g * s / (0.5 * period))
        c2 := b1
        c3 := -a1 * a1
        c1 := 1 - c2 - c3

    // Produce Nominal zero mean with zeros in the transfer response at DC and Nyquist with no spectral distortion
    zeros = (close - close[2]) 
    // SuperSmoother Filter
    filt := c1 * (zeros + zeros[1]) / 2 + c2 * nz(filt[1]) + c3 * nz(filt[2])
    
    // Compute Standard Deviation
    rms = math.sqrt(ta.ema(math.pow(filt, 2), period))
    // Rescale Filt in terms of Standard Deviations
    scaled_filt = rms != 0 ? filt / rms : 0
    alpha1 = math.abs(scaled_filt) * 5 / period
    dsma := alpha1 * close + (1 - alpha1) * nz(dsma[1])

    dsma

// @function Calculates trend percentage, color, and average DSMA
// @param src (series float) Source price series
// @param period (int) Initial calculation period
// @param step (int) Step size for increasing period
// @returns [float, float, color] Score, average DSMA, and color
percent_trend(src, period, step)=>
    dsma_arr = array.new<float>()

    // Calculate DSMAs with increasing periods
    length  = period
    dsma1   = dsma(src, length)
    length += step 
    dsma2   = dsma(src, length)
    length += step 
    dsma3   = dsma(src, length)
    length += step 
    dsma4   = dsma(src, length)
    length += step 
    dsma5   = dsma(src, length)
    length += step 
    dsma6   = dsma(src, length)
    length += step 
    dsma7   = dsma(src, length)
    length += step 
    dsma8   = dsma(src, length)

    // Store DSMAs in array
    dsma_arr.push(dsma1)
    dsma_arr.push(dsma2)
    dsma_arr.push(dsma3)
    dsma_arr.push(dsma4)
    dsma_arr.push(dsma5)
    dsma_arr.push(dsma6)
    dsma_arr.push(dsma7)
    dsma_arr.push(dsma8)

    val = 0.14285714285714285714285714285714

    // Calculate score based on DSMA comparisons
    score = 0.
    for i = 0 to dsma_arr.size() - 1
        dsma = dsma_arr.get(i)
        if dsma > dsma_arr.get(7)
            score += val

    // Determine color based on score
    color =  score > 0.5 
             ? color.from_gradient(score, 0.5, 1, na, upper_color) 
             : color.from_gradient(score, 0, 0.5, down_color, na)

    [score, dsma_arr.avg(), color]

// Calculate trend percentage, average DSMA, and color
[score, ma, color] = percent_trend(src, period, step)

up_percent = score * 100
dn_percent = 100 - up_percent

// ---------------------------------------------------------------------------------------------------------------------}
// 𝙑𝙄𝙎𝙐𝘼𝙇𝙄𝙕𝘼𝙏𝙄𝙊𝙉
// ---------------------------------------------------------------------------------------------------------------------{

// Creates a trend histogram visualization
if barstate.islast
    tbl = table.new(position.bottom_right, 100, 100)

    // Draw upper (bullish) part of histogram
    for i = 1 to math.round(up_percent/2)
        color_up = color.from_gradient(i, 0, 50, color.new(upper_color, 80), color.new(upper_color, 0))
        tbl.cell(0, 50-i, "", 1, 1, bgcolor = color_up)

        if i == math.round(up_percent/2)
            tbl.cell(0, 50-i, str.tostring(math.round(up_percent)), 1, 2, text_color = color_up, text_size = size.small)

    // Draw lower (bearish) part of histogram
    for i = 1 to math.round(dn_percent/2)
        color_dn = color.from_gradient(i, 0, 50, color.new(down_color, 80), color.new(down_color, 0))
        tbl.cell(1, 50-i, "", 1, 1, bgcolor = color_dn)

        if i == math.round(dn_percent/2)
            tbl.cell(1, 50-i, str.tostring(math.round(dn_percent)), 1, 2, text_color = color_dn, text_size = size.small)

// Detect crossovers for signal generation
cross_up = ta.crossover(score, 0.3)
cross_dn = ta.crossunder(score, 0.7)

// Plot signals
plotshape(cross_up ? ma : na, "", shape.diamond, location.absolute, color.new(upper_color, 50), size = size.small, force_overlay = true)
plotshape(cross_up ? ma : na, "", shape.diamond, location.absolute, upper_color, size = size.tiny, force_overlay = true)

plotshape(cross_dn ? ma : na, "", shape.diamond, location.absolute, color.new(down_color, 50), size = size.small, force_overlay = true)
plotshape(cross_dn ? ma : na, "", shape.diamond, location.absolute, down_color, size = size.tiny, force_overlay = true)

// Plot the main indicator line
plot(ma, color = color, linewidth = 2)

// ---------------------------------------------------------------------------------------------------------------------}
